
1. はじめに

　この度は toy-lang に関心をいただきましてありがとうございます。この文書は、プログラミング言

語 toy-lang について解説したものです。

　toy-lang は、個人的に開発している小さなスクリプト言語です。複雑な言語仕様や、処理速度を

求めることなく、プログラミング言語の機能や言語の実装について自分自身が知ることを目的とし

て実験的に開発しています。そのため、言語仕様については頻繁に変わる可能性があります。

　また toy-lang は、少なくとも現時点では実用的な言語ではありませんし、あくまでも個人の趣味

としての活動の産物のため、今まで仕様を整理してきませんでしたが、一度どのような言語であるか

ということを自分の中で整理する必要を感じ、本書を作成することにしました。

　色々と不備はあるかとは存じますが、どうぞ、お楽しみ頂ければと。

1. はじめに 4

2. 概要
2.1 特徴

　toy-lang は、以下のような特徴を持ったプログラミング言語です。

 インタプリタ言語

 シンプルな文法

 プロトタイプベースのオブジェクト指向

 C 言語による拡張性

 ファーストクラスとしてのクロージャ

 キーワード引数

 その他

(1) インタプリタ言語

　toy-lang はインタプリタ言語です。現在、いくつかの UNIX ライクな OS の上で動作します。

UNIX のスクリプト形式もしくは、toy-lang 組み込みのコマンドラインインタプリタで対話的に動作

させることができます。

(2) シンプルな文法

　toy-lang の全ての文法は、コマンド呼び出しもしくはメソッド呼び出しに引数を伴ったものとして

定義されます。一般的な言語にあるような文法やキーワードは存在しません。

例えば、以下は toy-lang の if 文です。

if {$i = 1} then: {do-something} else: {do-otherwise};

toy-lang では、if という構文が用意されているわけではありません。この文は、if コマンドの引数とし

て、コンディションを示す一パラメータである “{$i = 1}” 、コンディションが真の時に実行される

“then: {do-something}”キーワード引数、コンディションが偽の時に実行される“ else: {do-

otherwise}” キーワード引数、という 3 つのパラメータを持つコマンドの実行と考えることができま

す。

2. 概要 5

　各文の終端は“;”記号で示します。“;”が現れた時点で、その文を評価するという意味になりま

す。ただし、ブロック(ブロックについては後述しますが、“{}”で囲まれた文の集まりのこと)内の最後

の文については、“;”を省略することが可能です。

(3) プロトタイプベースのオブジェクト指向

　toy-lang は、プロトタイプベースのオブジェクト指向機能を持っています。オブジェクトを生成す

る際には、クラスを指定しますが、クラスの実態は、toy-lang のオブジェクトと等価です。新たに生

成されたオブジェクトは、生成時に指定したクラスにメッセージを委譲することでメソッドが選択され

実行されます。

　また、実行時に動的にクラスやオブジェクトに対してメソッドを定義することが可能です。

(4) C 言語による拡張性

　toy-lang は C 言語により記述されており、比較的簡単に C 言語による拡張が可能です。

　現時点では、実用的なプログラムを作成するために必要な機能はまだまだ足りないですが、必要

な機能については比較的簡単に追加することができると思います。

(5) ファーストクラスとしてのクロージャ

　toy-lang はクロージャ(実行時の環境を持つデータ型)を持ちます。クロージャの表現は、ソースス

クリプト上で“{}”で囲まれた部分です。また、クロージャは、ファーストクラスのデータ型であるため 、

関数の内外への持ち出しや変数への保持、任意の時点での評価が簡単にできます。

　クロージャは、実行時の環境を持つデータ型で、言語により意味や実装方法は様々ですが、toy-

lang の場合は、実行時のローカル変数およびカレントの実行オブジェクトを保持する構造です。

(6) キーワード引数

　toy-lang は、引数の表現が 2種類あります。ひとつは、多くのの言語で一般的な位置パラメータ

です。もうひとつは、オプショナルなパラメータを指定する際に便利なキーワードパラメータです。

　キーワードパラメータは、“name: value”のように、キーワードを名前 +“:”で指定し、その後に

渡したい値を指定します。また、キーワードパラメータは省略可能です。

2. 概要 6

(7) その他

　その他、リフレクション、例外処理、遅延評価の仕組みがあります。また、個人的に興味のある機能

なども随時追加されるかもしれません。

2.2 外観

　それでは、toy-lang という言語はどのような外観をしているでしょうか。簡単なサンプルプログラ

ムを以下に示します。

toy-lang の外観:
defun grep (pat file) {
    set f [new File];
    try {

$f open mode: i $file;
set n 1;
while {set r [$f gets]} do: {
    if {$r =~ $pat} then: {

print $file ":" $n ": " $r;
    };
    $n ++;
};

    }
    fin: {

$f close;
    };
}; 

　上記サンプルですが、UNIX の grep コマンドを真似た関数の定義です。この関数の起動の仕方

は、プロンプトに続いて以下のように入力します。

grep 'pattern' "file-name”;

　外観から読み取れる特徴としては、

1. “{” と“}” で囲まれた C 言語のような処理ブロック

2. while や if による制御構造

3. try による例外処理

4. “=~” による正規表現のパターンマッチング

5. よくわからない“;” による文の終端(!?)

6. UNIX シェルのような“$” 記号による変数の参照

7. C 言語っぽい演算子

などがあると思います。でも、ある程度の言語の経験のある方はそれほど違和感は無いのではない

かと思います。

2. 概要 7

2.3 影響を受けた言語

　toy-lang は、以下の言語の影響を受けています。

● C 言語

● Lisp (実は作者は本当はよく知らないのですが)

● Tcl

● Ruby

　言語の外観からすると、Tcl に一番近いかもしれません。これは、作者がしばらくの期間 Tcl のプ

ログラマであった影響によると思われます。内部的な動作や言語の基本的な機能は Lisp から来て

いる部分が多いでしょう。また、Ruby からは yield などのアイデアを借用しています。その他、作者

が過去に使用した言語から、知らずのうちに影響を受けているかもしれません。

2. 概要 8

3. 準備

3.1 プラットフォーム

　現時点で動作を確認しているプラットフォームは以下のとおりです。

 Ubuntu 8.04

 Ubuntu 9.04

 FreeBSD 7.1R

　実行形式を作るために、コンパイラは gcc、GC ライブラリとして Boehm GC および、正規表現ラ

イブラリとして鬼車が必要です。その他のライブラリとしては、UNIX の標準ライブラリのみですので、

一般的な UNIX 環境であればビルドは可能かと思います。

3. 準備 9

3.2 コンパイル方法

(1) 必要な外部ライブラリの用意

(1-1)  GC に  Boehm GC  ラ イ ブ ラ リ を使い ま す の で 、 以 下 の URL よ り バ ー ジ ョ ン

7.1alpha3-080224 を入手してインストールしてください。

http://www.hpl.hp.com/personal/Hans_Boehm/gc/gc_source/ 

(1-2) 正規表現ライブラリに鬼車を使わせていただきました。以下の URL よりバージョン 5.9.1 を

入手してインストールしてください。

http://www.geocities.jp/kosako3/oniguruma/index_ja.html

(2) ソースの入手 

以下の URL から、toy-lang の tar ball を入手します。

http://www31.atwiki.jp/toy-lang/pages/12.html

(3) ビルド

(3-1) 展開したディレクトリに入ります。

cd toy

(3-2) 最初に、Makefile の PREFIX を調整してください。

(3-3) make を実行します。

make

とすると実行ファイルができます。 

(3-4) インストールします。

make install

とすると、PREFIX にインストールされます。 

3. 準備 10

(4) テストを実行します。

cd test

./testall

とすると、簡単なテストを実行します。全てのテストで OK と出れば大丈夫ですが、リリースバージョ

ンによっては NG となる項目がある場合もあります。

3. 準備 11

3.3 処理系の構成

　toy-lnag を標準的な構成でインストールした際の処理系の構成について説明します。

3.3.1 ファイル構成

$PREFIX

/bin/toysh …　インタプリタ本体

/lib/toy/setup.toy …　起動時セットアップスクリプト

/lib/toy/lib/ …　ライブラリスクリプトディレクトリ

$HOME

/.toyrc …　ユーザ定義のセットアップスクリプト

　$PREFIX は、make 時に指定するインストール先ディレクトリです。

　$HOME は、ユーザのホームディレクトリです。

3.3.2 環境変数

　toy-lang では、以下の環境変数を参照します。

$HOME …　ユーザのホームディレクトリです。

3.3.3 グローバル変数

　toy-lang が起動したときにに定義されるグローバル変数です。

HOME …　ユーザのホームディレクトリです。

ENV …　環境変数の Hash オブジェクトです。

LIB_PATH …　ライブラリスクリプトが格納されたディレクトリのリストです。

ARGV …　インタプリタ起動時の引数リストです。

VERSION …　処理系のバージョンです。

CWD…　現在のカレントディレクトリです。

3. 準備 12

3.4 起動

　toy-lnag のインタプリタを起動するには、UNIX 上のシェルから、toysh コマンドを実行します。

インタプリタの起動:
$ toysh
*** Start toy-lang interpreter version 0.0.26.
> 

3.5 最新情報

　toy-lang についての最新情報は、今後も Webページで細々と発信してゆく予定です。以下の

URL にてご確認ください。

http://www31.atwiki.jp/toy-lang/

3. 準備 13

4. 言語仕様

4.1 データ型

　ここでは、toy-lang のデータ型について説明します。

　toy-lang の型は、以下の説明で特に断りの無い限り、そのほとんどがファーストクラスオブジェク

トとしての性質を持ちます。つまり、ほとんどの型が、変数への代入、引数への指定、戻り値としての

使用が可能です。

4.1.1 nil

　toy-lang で偽値を表します。ソース上では、“nil”で表現します。

　toy-lang では、nil 以外の全ての値は真として扱われます。

4.1.2 シンボル

　toy-lang のシンボルは、変数名、クラス名、メソッド名、ハッシュのキー値を表す名前です。名前に

使用可能な文字は以下のとおりです。

 A-Z、a-z

 0-9 (ただし、0-9 のみで構成される場合は、整数値とみなされます)

 ! % & - _ = ^ ~ | @ + * < > . / ?

上記の文字の一文字以上の組み合わせが、toy-lang で有効なシンボルとなります。

例: name

_123

my-name

4.1.3 参照

　シンボルの先頭に“$”をつけたものは、シンボルが示す値の参照となります。

例: $name

$?

文中に参照表現が現れた場合、その参照は、現在の環境の値(ローカル変数、オブジェクト変数

または、グローバル変数)に置換されてからコマンドもしくはメソッドが実行されます。

　参照は、厳密には(実行時の)型ではないため、従ってファーストクラスオブジェクトではありませ

ん(参照が現れた時点で実際のオブジェクトへ置換されてしまうため)。

4. 言語仕様 14

4.1.4 整数

　toy-lang の整数は、64bit符号付バイナリ値です。スクリプト内では 10進と 16進での表現が

可能です。

例: 123

0

-123

0x0000ffff

4.1.5 浮動小数点

　toy-lang の浮動小数点は、64bit 浮動小数点です。内部形式については、プラットフォームに依

存します。スクリプト内での表現は、仮数表示および指数表示の組み合わせにより行います。

例: .1

1.

.0

3.141592

-123.0

1E10

-1E-10

-.123E3

　指数表示がない場合は、仮数表示部に小数点(“.”)が必要となります。小数点が無い場合には、

整数と判断されます。ただし、“.” のみの場合、浮動小数点ではなく、シンボルとして判断されます。

また、指数表示が存在する場合は小数点は不要です。

4.1.6 文字列

　toy-lang の文字列は、ダブルクォート“"”で囲まれた文字の列です。また、文字列内では、エス

ケープ記号( \ )で特定の文字コードを表現することが可能です。

例: "Hello World"

""

"End\n"

　文字列内で使用可能なエスケープ記号は、以下のとおりです。

4. 言語仕様 15

 \\  →エスケープ記号

 \t  →タブ

 \n  →ラインフィード

 \r  →キャリッジリターン

 \"  →ダブルクオート

4.1.7 正規表現文字列

　toy-lang では正規表現パターンを表すための型を用意しました。正規表現文字列は、文字列と

似ていますが、シングルクォート“'”で囲まれた文字列であるところが異なります。また、利用可能な

エスケープ記号も異なります。

例: '[A-z0-9].*'

'\(.*\)'

　文字列内で使用可能なエスケープ記号は、以下のとおりです。

 \\  →エスケープ記号

 \'  →シングルクォート

文字列とは異なり、上記以外の組み合わせ以外で単独で現れたエスケープ記号は、エスケープ

記号そのものを表します。

4.1.8 リスト

　toy-lang のリストは、“(” と “)” で囲まれたデータの列です。要素間は空白文字(スペース、タ

ブおよび、改行)により区切ります。また、各要素は任意のデータ型を指定できます。もちろん、リスト

の中の要素としてリストを指定することも可能です。

　また、Lisp 処理系におけるドット対の表記も可能です。この際は、Lisp と同様 cons セルの car

部、cdr 部のそれぞれの要素を指定することが可能です。

例: ()

(a b c)

("a" b 0 (1 2 3))

("a" . {do-something})

4. 言語仕様 16

4.1.9 クロージャ(ブロック)

　クロージャを説明する前に、まず、toy-lang のブロックについて説明します。ブロックとは、“{” と 

“}” で囲まれた文の集まりです。通常の使い方としては、例えばループの処理部や関数の本体な

どの一連の処理のかたまりを表すために使用します。

ブロックの例:
set i 0;　
while {$i < 10} do: {　

    # ループ本体のブロック　
    $i ++;　
};

　スクリプト上に現れる “{” と “}” で囲まれた部分は全てブロックです。もちろん、ブロックの入れ

子も可能です。

　これだけだと、見た目では C 言語のブロックとはあまり違いはありません。

　もうひとつ説明すると、toy-lang のブロックは、静的な構造であるということです。それでは、ブロ

ックに対応する動的な構造とはなんでしょうか。それがクロージャです。

　クロージャは、スクリプトが実行され、ブロックが含まれる文に処理が到達したときに生成される

データ型です。クロージャを構成するデータ型は、ブロックが現れたときの実行環境とそのそのブロ

ック自身を含みます。toy-lang のクロージャが持つ実行環境とは、クロージャが生成されたときのロ

ーカル変数とインスタンス変数です。

　クロージャを使うと、処理のかたまりと環境のセットを簡単に受け渡したり、別の環境であとから実

行したりといったことができます。

クロージャによるブロックの持ち出しの例:
defun foo (x) {return {println $x " world"}};　
set x [foo "hello"];　
println $x #  {println $x " world"}→ 　
$x eval; #  hello world→

4. 言語仕様 17

4.1.10 評価ブロック

　toy-lang は、文の評価を明示的に示す必要があります。評価ブロックは、“[” と “]” で囲まれた

文の集まりです。表記自体はブロックと似ています。スクリプト中で、評価ブロックを含む文に到達し

たとき、そのコマンドやメソッドの実行に先立ち “[” と “]” とで囲まれたスクリプトが実行され、評

価ブロックが記述された部分にその値が埋め込まれます。評価ブロックの実行環境は、実行時の環

境のままです。従って、評価ブロックの中では、評価ブロックの外と同じ変数が参照可能です。

　Lisp では、評価は自動的に実行されますが、toy-lang の場合は、プログラマが明示的に示す必

要があることが Lisp と大きく異なる部分です。

評価ブロックによる置換の例:
set i 10;　
println [$i ++]; #  11→

　評価ブロックは、厳密には(実行時の)型ではないため、従ってファーストクラスオブジェクトでは

ありません(評価ブロックが現れた時点で文の評価後のオブジェクトへ置換されてしまうため)。

4.1.11 関数

　toy-lang の関数もまたファーストクラスオブジェクトの性質をもちます。関数は、 fun コマンド、

defun コマンドおよび、Object クラスの method メソッドにより生成されます。

　fun コマンド、defun コマンドおよび method メソッドにより生成されるのはすべて関数型の型で

あり、生成のされかたによりその関数がどのように管理されるかが異なります。

　fun コマンドにより、名前の無い関数を生成することが可能です。

　defun コマンドにより、名前付きの関数を生成し、グローバルなスコープで関数を名前で呼び出

すことができるようになります。

　method メソッドにより、オブジェクトおよびクラスに対してオブジェクトのメンバ関数を定義する

ことができます。オブジェクのメソッド呼び出しによりオブジェクトのメンバ関数を呼び出すことがで

きるようになります。

4.1.12 オブジェクト

　toy-lang のオブジェクトは、new コマンドにより生成されます。オブジェクトは、その構成要素とし

て、委譲先のクラス、メンバ変数、メンバ関数を含みます。

　オブジェクトに対して委譲先のクラスを複数指定することができ、これにより多重継承の仕組み

を実現しています。

4.1.13 制御

　toy-lang の制御型は、関数からの戻り値を指定して関数を終了したり、ループの実行を中断、再

開したりするためのものです。

　制御は、以下のコマンドにより生成されます。

● return

● break

● continue

● retry

● redo

　return は、関数の実行を中断し、return の引き数の値を結果として呼び出し元へ返します。

　break は、ループの処理を中断します。値を指定することにより、結果を返すことが可能です。

　continue は、現在のループの処理を中断し、次の要素の処理を開始します。

　retry は、現在のループの処理を中断し、ループの最初から処理をやり直します。

　redo は、現在のループの処理を中断し、もう一度同じ要素の処理を再開します。

4.1.14 例外

　例外は、toy-lnag の処理系の様々な部分で発生します。発生の原因は、スクリプトの記述間違い、

引数の型の不一致、スクリプト中での明示的記述などで、たくさんの要因があります。

　例外が発生すると、関数の呼び出しを遡って例外が伝播されます。例外を補足するためには、try

コマンドを使用します。関数の呼び出しを遡る途中で、try コマンドにより例外が補足されない場合は

トップレベルまで伝播し、最終的にインタプリタによりエラーが報告されます。

　try コマンドが例外を補足した際に、発生した例外を参照するには、 try コマンドの catch:ブロック

に渡されるバインド変数を使用する必要があります。

　スクリプト中で例外を明示的に発生させるためには、throw コマンドを使用します。

4.1.15 バインドリスト

　バインドリストとは、イテレータを構成するブロックに渡されるバインド変数のリストです。バンドリ

ストは、“|” と “|” とで囲まれたシンボルのリストです。以下は、バインドリストの使用例です。

バインドリストの例:

(1 2 3) each do: {| i | # | i | はバインドリスト

　　　　println $i # iはンドバイ変数 
};

　toy-lan の文は、基本的に “;” により終端しますが、バインドリストに関しては例外的に “;” は

不要となっています。

注意事項　“|” 記号は、シンボル名の一部としても使用可能です。従って、

バインドリストを記述する際の “|” の前後には空白文字が必要です。空白が

無い場合、前後の文字と合わせてシンボルと判断されてしまいます。

4.2 クラス

　toy-lang のクラス構成は以下のとおりです。

Object 全てのオブジェクトの基底クラスです。

Integer 整数型の wrapper クラスです。

Real 浮動小数点型の wrapper クラスです。

List リスト型の wrapper クラスです。

String 文字列型の wrapper クラスです。

RQuote 正規表現型の wrapper クラスです。

Block ブロック(クロージャ)型の wrapper クラスです。

Hash Hash クラスです。キー・値のペアでデータを管理するクラスです。

Array Array クラスです。任意の型をインデックス番号で管理するクラスです。

Functions 名前付き関数が便宜的にメンバとして登録されるクラスです。

File ファイル入出力のためのクラスです。

　toy-lang における wrapper クラスは、基本的なデータ型(整数、浮動小数点、リスト、文字列、正

規表現文字列および、ブロック)に対してクラスとしての性質を付与するためのものです。ソースコー

ド中に、これらの基本的なデータ型がオブジェクトの位置に現れた際には、自動的に対応する

wrapper クラスで wrap されたオブジェクトが生成され、wrapper クラス内のメンバ関数が呼び出

されます。

　各クラスの詳細については、 「5章　リファレンス」を参照して下さい。

4.3 プログラムの構成

　toy-lang のプログラムはスクリプトと文により構成されます。以下の例は、小さなスクリプトと文の

例です。ひとつのスクリプトに二つの文が含まれています。

プログラムの例:
defun add1 (x) {
    fun () {$x ++ 1};
};
[add1 100]; #  101→

(1) スクリプト

　スクリプトは、toy-lang で記述されたプログラムです。スクリプトは、通常ファイルにテキスト形式

で記述し、load コマンドによりインタプリタに読み込まれ、評価されます。

スクリプトファイルの拡張子は、“.toy” です。

　また、スクリプトは、toy-lang の文の集まりです。先に説明したブロックや評価ブロックの中も文の

集まりであり、再帰的にスクリプトが現れます。

(2) 文

　文は、toy-lang の評価の最小単位であり、スクリプト中では文が先頭から順に実行されることで

プログラムのアルゴリズムを実際に実現してゆきます。スクリプト中の各文は、“;”で終端します。

(3) コメント

　toy-lang のコメントは、“#” で始まり、その行の終わりまでとなります。

コメントの例:

# この行はコメントです。

set i 0; # シャープ記号(#)から行の最後までがコメントです。

(4) スクリプト/関数の値

　スクリプトの値は、スクリプトの最後の文が実行された結果となります。また、“result” コマンド

で明示的に示すことも可能です。文がイテレータブロックの場合は、“break” コマンドの引数も値

となります。

　さらに、関数のスクリプトを実行した場合は、“return” コマンドにより戻り値を指定することも可

能です。

(5) 文の実行結果

　文を実行した後は、特殊なローカル変数である、“?”  にその結果が設定されます。参照する際に

は、“$?”  とします。

文の実行結果:
set i 100;
$i + 1;
println $?; # → 101

4.4 構文規則

　toy-lang の文は、以下の規則に従い評価されます。

● 関数　[ 引数　... ]　; …　構文規則 1

● オブジェクト　メソッド　[ 引数　... ]　; …　構文規則 2

(1) 関数呼び出し

　構文規則 1 は、組み込み関数、ユーザ定義関数および、カレントオブジェクトのメソッドを呼び出

すための構文です。

　文の最初のキーワードが評価され、組み込み関数、ユーザ定義関数もしくは、メソッドである場合

に、その関数が実行されます。

関数呼び出しの例:

set x 100; # → 組み込み関数の呼び出し

foo ”a string”; # → ユーザ定義関数の呼び出し

　オブジェクトに定義したメソッドの呼び出しに関して、同一オブジェクトのコンテキストでは、オブ

ジェクトの指定は省略可能であり、その際は、関数呼び出しの形式で自オブジェクトのメソッドが起

動されます。以下は、暗黙のメソッド呼び出しの例です。

暗黙のメソッド呼び出しの例:
class F;
F method m1 (x) {

    m2 $X; # F::m2 の呼び出し
};
F method m2 (x) {
   println $x;
};

(2) メソッド呼び出し

　構文規則 2 は、オブジェクトに対してメソッドを適用するための構文です。

　文の最初のキーワードが評価され、それがオブジェクトの場合、次のキーワードを評価し、オブジ

ェクトのメソッドの探索が行われます。メソッドが見つかった場合、そのオブジェクトの環境が作られ、

メソッドが実行されます。

メソッド呼び出しの例:
class F;
F method m1 (x) {
    println $x;
};
set o [new F];
$o 100; # → 100

　この呼び出し規則は、組み込みのデータ型でも良く使用します。例えば、整数型に対する演算な

どは、この呼び出し規則の典型的な例となります。

整数データによる Integer クラスメソッドの呼び出しの例:
set i 0;

$i + 1; # Integer::+ の呼び出し

(3) クラスメソッド呼び出し

　構文規則 2 のもうひとつの場合として、文の最初のキーワードがクラスの場合、そのクラスのメソ

ッドが呼び出されます。

クラスメソッドの呼び出しの例:
Integer method foo　()　{
    # do something
    ...
};

Integer foo; # クラスメソッドの呼び出し

　この例のように、組み込みクラスおよび、ユーザ定義のクラスに対して、動的にメソッドを定義する

ことが可能です。

4.5 変数

　toy-lang には、プログラム中で設定したり参照することが可能な変数があります。変数のクラスと

しては、ローカル変数、インスタンス変数および、グローバル変数があります。

　変数の参照は、文中で、変数のシンボル名の先頭に“$” を付与することにより行います。プログ

ラムのコンテキストで、各変数クラスに同名の変数が存在した場合、

ローカル変数　→　インスタンス変数　→　グローバル変数

の順序で変数の検索が行われます。

　toy-lang の変数は動的なものです。スクリプトの任意の時点で任意の型のデータを設定するこ

とが可能です。

(1) ローカル変数

　ローカル変数は、関数の呼び出し毎にスタックに生成される環境です。

　ローカル変数スロットは、実行中の関数内で参照可能なシンボルの辞書構造です。ローカル変

数の参照の際は、スタックポインタにあるローカル変数スロットからシンボルの値が参照されます。

　関数の呼び出しにより、新たなローカル変数スロットが生成され、関数の終了とともにそのスロッ

トは消滅します(正確には、関数内にてクロージャが生成された際には、クロージャの構成の一部と

して保持されます)。

　ローカル変数の設定には、“set” コマンドを使用します。また、ローカル変数がローカル変数スロ

ットに存在するかを調べるには、“set?” コマンドを使用します。

(2) インスタンス変数

　インスタンス変数は、オブジェクトが保持する変数スロットで、オブジェクトのメソッド呼び出しに

伴い生成される環境です。メソッド呼び出しにあたっては、オブジェクトスタックとローカル変数スタ

ックが生成されます。

　toy-lang には、ローカル変数スタックのほかに、オブジェクトスタックというものもあり、オブジェ

クトのメソッド呼び出しに伴い、ローカル変数スタックに加えてオブジェクトスタック上にも新たな環

境が生成されます。オブジェクトスタック上のインスタンス変数スロットは、オブジェクトが持つイン

スタンス変数スロットへの参照になっています。

　インスタンス変数の設定には、“sets” コマンドを使用します。また、インスタンス変数がインスタ

ンス変数スロットに存在するかを調べるには、“sets?” コマンドを使用します。

(3) グローバル変数

　toy-lang のグローバル変数は、スクリプト中の全ての場所からの設定、変更が可能です。

　グローバル変数の定義には、“defvar” コマンドを使用します。また、内容の変更には、

“setvar”  コマンドを使用します。二重にグローバル変数を定義しようとすると、例外が発生します。

グローバル変数が存在するかを調べるには、“defvar?”  コマンドを使用します。

4.6 関数

　toy-lang の関数は、名前なし関数と名前つき関数があります。関数の実態は、名前なし関数であ

り、先に説明した関数データ型のデータがその本体です。名前つき関数は、名前なし関数が

Functions クラスのインスタンス変数に登録されたものです。

　名前なし関数の生成は、“fun” コマンドを使用します。名前なし関数の呼び出しは、変数の参照

や、　関数からの戻り値を実行するなどの方法があります。

名前なし関数呼び出しの例:
set foo [fun (i) {$i + 1}];

$foo 100; # 名前なし関数の呼び出し

　名前つき関数の定義は、“defun” コマンドを使用します。呼び出しは、関数名を指定することに

より行います。

名前つき関数呼び出しの例:
defun foo (i) {$i + 1};

foo 100; # 名前つき関数の呼び出し

4.7 クラス

　toy-lang のクラスは、オブジェクトをクラスの辞書へ名前をつけて登録したものに過ぎません。ク

ラスという言葉は、便宜的なものであり、実態はオブジェクトそのものです。

　クラス(としてのオブジェクト)の目的は、他のクラスを定義する際のテンプレートとして名前を参照

するためと、オブジェクトを生成するためのデレゲートクラスを指定するためにあります。

　クラスは、“class” コマンドで生成します。本コマンドでクラスを定義する際に、デレゲートクラス

を指定することができます。

　デレゲートクラスとは、普通の言葉で言うところの親クラスのことで、 toy-lang の場合はこれを複

数指定することが可能であり、多重継承のようなことも可能です。

クラス定義の例:

# 単純なクラスを定義(デフォルトで Object クラスがデレゲートクラスとなる)
class bar;

# クラスfooを定義(bar, baz クラスにデレゲートする)
class foo bar delegate: (baz);

4.8 オブジェクト

　toy-lang は、標準で簡単なオブジェクトシステムをサポートしています。オブジェクトの意味合い

は、他の言語と似ていて、オブジェクトとそれに包括されるメンバ関数、メンバ変数を含むデータの

集合です。また、オブジェクトはデレゲートクラスを持ち、継承と似た動作も行えます。

　toy-lang のオブジェクトは動的なもので、実行中にメソッドの定義が可能であり、mix-in のような

事を簡単に実現できます。

　オブジェクトは、“new” コマンドにより生成されます。オプションとして既定のクラスを指定するこ

とにより、クラスからオブジェクトを生成する仕組みと似た動作をさせることも可能です。

オブジェクト生成の例:

# 単純なオブジェクトの生成(デフォルトでは Object クラスのインスタンスと

なる)
set o [new];

# クラスを指定してオブジェクトを生成
set o [new foo];

4.9 メソッド

　toy-lang のメソッドは、オブジェクトのメンバとして登録された関数のことです。メソッドは、クラス

およびオブジェクトに対して定義することができます。

　メソッドの定義は、メソッドを定義したいクラスもしくはオブジェクトに対して “method” メソッド

を呼び出すことにより行います。ちなみに “method” メソッド自身は、全ての基底オブジェクトであ

る Object クラスに定義されたメソッドです。

メソッド定義の例:

# ユーザ定義クラスにメソッドを定義する
class MyClass;
MyClass method foo(x) {println $x};

# オブジェクトに対してメソッドを定義する
set x [new MyClass];
$x method bar(x) {println $x};

# wrapper 組み込みのクラスに独自のメソッドを定義する
List method print () {println [self]};
(a b c) print; # → ”(a b c)”

　メソッドの呼び出しは、オブジェクトにメソッド名を指定することにより行うか、もしくは、そのオブジ

ェクトのメソッドが実行中の場合は、直接メソッド名を指定することにより行います。

メソッド呼び出しの例:

# クラス/メソッドを定義し、オブジェクトを生成
class MyClass;
MyClass method foo(x) {println $x};
set x [new MyClass];

# メソッドの呼び出し
$x foo ”abc”; # → ”abc”

4. 言語仕様 30



Toy-lang Language Manual

4.10 引数

　toy-lang では他の言語と同様に、関数やメソッドを定義する際に引数を定義することができます 。

関数やメソッド定義に指定する引数のことを仮引数と言います。仮引数定義は、関数名、メソッド名、

もしくは fun コマンドの場合には fun コマンドの直後に、“(” と“)”で囲まれた部分に記述します。

　また、関数やメソッドを呼び出す際には、その関数やメソッドに与えるための引数を指定しますが、

これを実引数と呼びます。実引数の指定は、関数名やメソッド名に続いて、関数に与えたい値を列

挙することにより行います。

仮引数と実引数の例:

# 関数定義における仮引数

defun foo (x y z) {...}; # x, y, z が仮引数

# 関数呼び出しにおける実引数

foo 1 ”abc” (a b c); # 1, ”abc”, (a b c) が実引数

　toy-lang での引数の指定の方法にはいくつかのルールがあります。

1. 位置引数

2. キーワード引数

3. スイッチ引数

4. 可変長引数

5. 遅延評価引数

上記の各引数のそれぞれは、ひとつの関数およびメソッドの定義内で組み合わせて指定することが

可能です。以下、それぞれの引数について説明します。

(1) 位置引数

　位置引数とは、その名の通り仮引数と実引数の対応を引数の位置により指定するものです。仮

引数部分では、引数のシンボル名をリストの形で順に列挙します。実引数として渡す際には、関数

名、メソッド名に続いて関数に渡す値を順に列挙します。位置引数の場合、仮引数と実引数の数は

一致しなければなりません。

位置引数の例:

# 位置引数の定義

defun foo (x y z) {...}; # x, y, z が仮引数

# 位置引数を指定して関数を呼び出す

foo 1 ”abc” (a b c); # 1, ”abc”, (a b c) が実引数

上記の例の場合、関数 foo が呼び出された際には、x に 1 が、y に "abc" が、そして、z にリストとし

て (a b c) がそれぞれ渡されます。

(2) キーワード引数

　キーワード引数とは、呼び出し側と関数定義側との引数の受け渡しをキーワードにより指定するも

のです。位置引数とは異なり、呼び出し時には仮引数側で指定されたすべてのキーワード引数を指

定する必要はありません。呼び出し時には、必要な引数のみ指定することが可能です。また、指定の

順序についても制限はなく、呼び出し側で任意の順序でキーワード引数を指定することができます。

　キーワード引数の定義は、仮引数側の定義は、 「キーワード: シンボル名」となります。また呼

び出し時の指定は、 「キーワード: 値」となります。仮引数側のキーワードと呼び出し側のキーワー

ドが一致した際に、呼び出し時の値がキーワード仮引数のシンボル名に設定されます。

　呼び出し時にキーワード引数が指定されなかった場合は、関数側のシンボルは設定されません。

そのため関数側では、キーワード引数が与えられているかどうかを知るために、

“set?”  コマンドを使って調べる必要があります。

キーワード引数の例:

# キーワード引数の定義
defun if (cond then: then-body else: else-body) {...};

# キーワード引数を指定して関数を呼び出す

if $x then: {set i 0} # then: キーワード引数を指定、else: は未設定

上記の例は、if コマンドを模擬的に示したものです。関数呼び出しにおいて thne: キーワードを指定

した場合、if コマンドが呼び出された際には、シンボル名 then-body に値(ここでは、{set i 0} )が

設定されます。else: キーワードは設定されていないため、else-body シンボルには値は設定されま

せん。また、この例では、仮引数の定義で、第 1 引数には cond というシンボルが指定されています

が、これは、位置引数とキーワード引数を混在して指定可能であることを示しています。

(3) スイッチ引数

　スイッチ引数は、キーワード引数の実引数を指定する際の糖衣構文です。キーワード引数の実引

数を指定する際に「:キーワード」とすることにより、値を省略することができます。仮引数には、非

nil値が設定されます。スイッチ引数の目的は、パラメータの有無のみが意味を持つような引数の与

え方を簡単にすることです。

スイッチ引数の例:

# キーワード引数の定義
defun foo (nocase: case-switch) {...};

# スイッチ引数を指定して関数を呼び出す

foo :switch; # foo の case-switch には非 nilが設定される

foo; # foo の case-switch には値が設定されない

(4) 可変長引数

　位置引数は、仮引数と実引数の数が一致する必要があります。実引数として任意個数の引数を

渡したい場合は、キーワードパラメータとして「args: シンボル名」を指定します。このようにすると、

位置引数より多い実引数の残りすべては、関数が呼び出された際に args: キーワード引数のシン

ボル名にリストとして設定されます。

可変長引数の例:

# 可変長引数を利用した関数の呼び出し
foo 1 2 3;

# 可変長引数の定義と動作
defun foo (a args: rest-parameters) {
  println $a; # → 1
  println $rest-parameters # → (2 3)
};

　可変長引数と位置引数を同時に指定する際には、実引数の個数は最低限仮引数で定義した個

数が必要です。

(5) 遅延評価引数

　遅延評価引数は、クロージャブロックの評価を仮引数の参照時まで遅延するための機能です。遅

延評価引数を指定するには、関数定義の仮引数シンボル名を「&シンボル名」と記述することによ

り行います。呼び出し側の実引数は、遅延評価引数に対してクロージャブロックを指定します。

遅延評価引数の例:

# 遅延評価引数の定義
defun foo (&a) {

  println $a; # → 1 (初回の参照なのでクロージャ{$i ++}が評価される)

  println $a; # → 1 (2回目以降の参照は、最初の評価結果となる)
};

# 遅延評価引数を用いた関数の呼び出し
set i 0;
foo {$i ++};

　遅延評価引数にクロージャブロックを渡すと、関数内で、最初にそのシンボルを参照した際にクロ

ージャが実行されます。クロージャの実行環境は、実引数の環境となります。遅延評価引数が 2 度

目以降に参照された場合は、最初のクロージャの実行結果(スクリプトの最後の値)が記憶されます

ので、その記憶された値となります。複数回クロージャが実行されることはありません。

　遅延評価引数に対して、実引数としてクロージャブロック以外を渡した場合は、通常の変数の参

照と同等になります。

4.11 マクロ

　toy-lang には、パーサに組み込みのマクロ的な構文があります。Lisp のようにユーザがマクロを

作成することはできません。

4.11.1 get マクロ

　get マクロは、Object に対して get メソッドを適用するための糖衣構文です。もともとは、Hash の

要素を参照する際の簡易な構文として考えましたが、1 引数の get メソッドを持つオブジェクトに対

してはどのオブジェクトに対しても使用可能です。

　get マクロの記法は、文の中で以下のように記述します;

オブジェクト,パラメータ

このように記述した際、実行時には以下のように置換され、オブジェクトの get メソッドが呼び出さ

れます。

[オブジェクト get パラメータ]

get マクロの例:

# Hashの生成と要素の代入
set o [new Hash];
$o set ”foo” 1;

# 要素の参照(通常の記法)
println [$o get ”foo”]; # → 1

# 要素の参照(get マクロによる記法)
println $o,”foo”; # → 1

4.11.2 init マクロ

　init マクロは、オブジェクトの生成の際に実行されるコンストラクタ(クラスの init メソッド)を簡易

に呼び出すための糖衣構文です。

　init マクロの記法は、文の中で以下のように記述します;

`クラス名 パラメータ

このように記述した際、実行時には以下のように置換され、クラスの init メソッドが呼び出されます。

[new クラス名 arg: (パラメータ)]

init マクロの例:

# クラス定義
class Foo;
Foo method init (a) {....};

# オブジェクトの生成(通常)
set o [new Foo arg: (1)];

# オブジェクトの生成(init マクロによる記法)
set o `Foo 1;

4.12 関数のオートロード

　toy-lang には、関数のオートロード機能があります。

　未知の関数名が呼び出された際にインタプリタはグローバル変数 LIB_PAHT に記述されたサー

チパスの順に関数が定義されたファイルを探し、自動的にスクリプトファイルをロードし、その後関数

の呼び出しを行います。

　スクリプトファイルの命名規則は、 「関数名 + ”.toy”」です。本スクリプトファイル中には、当該

関数の定義(defun)がある必要があります。もしスクリプトファイル中に当該関数の定義が無い場

合、関数呼び出し毎に毎回ファイルがロードされることになります。

5. リファレンス
5.1 コマンドリファレンス

5.2 クラスリファレンス

